TEMPLATE = app
TARGET = edgygui
DEPENDPATH += edgy edgy/yaml .
INCLUDEPATH += edgy edgy/yaml .
OBJECTS_DIR = obj

# Input
HEADERS += \
	conf.h\
	customwidgets.h\
	edgyconfig.h\
	cursorselect.h\
	keycapture.h\
	globals.h\
	edgy/yaml/*.h
SOURCES += \
	edgy/*.cc\
	edgyconfig.cpp\
	main.cpp\
	cursorselect.cpp\
	conf.cpp\
	customwidgets.cpp\
	keycapture.cpp\
	edgy/yaml/*.c
FORMS += edgyconfig.ui cursorselect.ui keycapture.ui

DEFINES = HAVE_CONFIG_H NO_BINARY_EDGY

LIBS += -lXcursor -lXtst -lstdc++
CONFIG += qt debug warn_on x11
# Comment the following if need debug info
CONFIG += release

# Variables

# INSTALLPREFIX *= /usr/local
# the above is buggy, so use the isEmpty() function below
isEmpty(INSTALLPREFIX) {
	INSTALLPREFIX = /usr/local
}
BINPREFIX *= $${INSTALLPREFIX}/bin
DOCPREFIX *= $${INSTALLPREFIX}/share/doc/edgygui

# Installs
binary.path = $${BINPREFIX}
binary.files = $${TARGET}

documentation.path = $${DOCPREFIX}
documentation.files = docs/*

# this has to come after the install defs
INSTALLS += binary documentation
